Solutions to Exercise: Diffusion

PF@SuS@UniHD 2016+

0. Constants and Default Values

Space and Time units in this section are μm and ns!

```
In[77]:= Uth = Ck T / Cq;

In[78]:= DEFAULTS = {d → 300, VDep → 100,
    T → 300, μe → 128, Ecrit → 0.7,
    Ck → 1.3806503 × 10^-23, Cq → 1.60218 × 10^-19};
  (*Ck in J/K = V C/K, μe in um^2/(V ns),
  μe and Ecrit for electrons*)

In[79]:= SetOptions[Plot,
   {Frame → True, Filling → Axis}];
```

1. Field E(x)

```
In[80]:= Clear[VDep, d, V, Field, Tdrift, σ]

In[81]:= Field[x_, V_] =
   2 VDep/d (d - x)/d + (V - VDep)/d // Simplify
```

Out[81]=

$$\frac{d\,(V + VDep) - 2\,VDep\,x}{d^2}$$

In[82]:=
```
V == ∫_0^d Field[x, V] ⅆx (* Check that integral is ok *)
```

Out[82]=
```
True
```

In[83]:=
```
FieldAverage[V_] = V / d;
(* we need this later *)
```

In[85]:=
```
Manipulate[
 Show[Plot[Evaluate[{Field[x , V],
      Field[x , VDep], FieldAverage[V]} /.
     {VDep → 100,d → 300}], {x, 0, 300},
   PlotRange → {0, 1}, ImageSize → Medium]]
 , {{V, 150, "Bias"}, 100, 200,
  Appearance → "Labeled"}]
```

Out[85]=



2. Drift time

In[86]:=
```
vconst[x_, V_] = μe Field[x, V]
  (* assume constant mobility *)
```

Out[86]=
$$\frac{(d\ (V + VDep) - 2\ VDep\ x)\ \mu e}{d^2}$$

In[87]:=
```
0 == vconst[d, VDep]
 (* check that drift speed becomes
  zero at d just at depletion *)
```

Out[87]= True

In[88]:=
```
$Assumptions = d > 0 && start > 0 &&
   start < d && V > 0 && VDep > 0 && V > VDep;
```

v(x) = dx/dt -> dt = dx / v(x)

In[89]:=
```
DriftTimeConst[V_, start_] =
```
$$\int_0^{start} \frac{dx}{vconst[x, V]}\ \text{// Simplify}$$

Out[89]=
$$\frac{d^2\ \mathrm{Log}\left[\frac{d\ (V+VDep)}{-2\ start\ VDep+d\ (V+VDep)}\right]}{2\ VDep\ \mu e}$$

In[90]:=
```
Plot[Table[DriftTimeConst[V, x],
    {V, 120, 200, 40}] /. DEFAULTS,
 {x, 0, 300}, PlotRange → Full, FrameLabel →
  {"Start Depth [μm]", "Drift Time [ns]"}]
(* Plot drift time as a function
 of the start position. Check
 different voltages *)
```

Out[90]=



In[91]:=
```
DriftTimeConst[200, 300] /. DEFAULTS //
 N (* Typical case:
   200V and start at the other side *)
```

Out[91]= 3.86231

3. Case for V = VDep

In[92]:=
```
DriftTimeConst[VDep, start] // Simplify
 (* See what happens just at depletion *)
```

Out[92]= $$\frac{d^2 \, \text{Log}\left[\frac{d}{d-\text{start}}\right]}{2 \, \text{VDep} \, \mu e}$$

In[93]:=
```
Limit[%, start → d] (* check that,
just at depletion,
charges never arrive when starting at d*)
```

Out[93]= $$\frac{\infty}{\mu e}$$

4. Compare to Average Field Case

In[94]:=
```
Clear[vav]; vav[V_] = μe FieldAverage[V]
```

Out[94]= $$\frac{V \, \mu e}{d}$$

In[95]:=
```
Clear[DriftTimeAverage]; DriftTimeAverage[V_, x_] = x / vav[V]
```

Out[95]= $$\frac{d \, x}{V \, \mu e}$$

In[96]:=
```
DriftTimeAverage[V, x] / DriftTimeConst[V, x]
```

Out[96]= $$\frac{2 \, \text{VDep} \, x}{d \, V \, \text{Log}\left[\frac{d \, (V+\text{VDep})}{d \, (V+\text{VDep}) - 2 \, \text{VDep} \, x}\right]}$$

In[97]:=
```
Plot[{DriftTimeAverage[V, x],
    DriftTimeConst[V, x]} /. V → 105 /.
   DEFAULTS, {x, 0, 300},
  PlotRange → Full, FrameLabel →
   {"Start Depth [μm]", "Drift Time [ns]"}]
(* When we start close to the
  junction (small x),
real case is faster because field
  is high. For high voltages,
triangular field effect is
  reduced and curves get closer *)
```

Out[97]=



5. Sigma

In[98]:=
```
Clear[ChargeSigma];
ChargeSigma[V_] = Sqrt[2 μe Uth DriftTimeConst[V, d]] // Simplify
```

Out[98]=
$$\sqrt{\frac{Ck\, d^2\, T\, Log\left[\frac{V+VDep}{V-VDep}\right]}{Cq\, VDep}}$$

In[99]:=
```
AverageSigma[V_] = Sqrt[2 μe Uth DriftTimeAverage[V, d]] // Simplify
```

Out[99]= $\sqrt{2}\, d \sqrt{\frac{Ck\, T}{Cq\, V}}$

In[100]:=
```
Plot[{ChargeSigma[V], AverageSigma[V]} /. DEFAULTS, {V, 101, 200},
 PlotRange → {0, 10}, FrameLabel → {"Voltage [V]", "Charge cloud Sigma [μm]"}]
(* Real sigma is a bit larger than simple average model
 (longer drift). Difference is smaller for high bias. *)
```

Out[100]=



6. Unused